

FizzBuzz in Haskell by Embedding a Domain-Specific Language

by Maciej Piróg <maciej.adam.pirog@gmail.com>

April 23, 2014

The FizzBuzz problem is simple but not trivial, which makes it a popular puzzle during job interviews for software developers. The conundrum lies in a peculiar but not unusual control-flow scenario: the default action is executed only if some previous actions were not executed. In this tutorial, we ask if we can accomplish this without having to check the conditions for the previous actions twice; in other words, if we can make the control flow follow the information flow without losing modularity. The goal is to have the most beautiful code!

We deliver a rather non-standard, and a bit tongue-in-cheek solution. First, we design a drastically simple domain-specific language (DSL), which we call, after the three commands of the language, Skip-Halt-Print. For each natural number n , we devise a Skip-Halt-Print program that solves FizzBuzz for n . Then, we implement this in Haskell, and, through a couple of simple transformations, we obtain the final program. The corollary is a reminder of the importance of higher-order functions in every functional programmer's toolbox.

The FizzBuzz problem

FizzBuzz is a simple game for children, and therefore a really hard nut to crack for programmers and computer scientists. To quote the rules [1]:

Players generally sit in a circle. The player designated to go first says the number '1', and each player thenceforth counts one number in turn. However, any number divisible by three is replaced by the word *fizz* and any divisible by five by the word *buzz*. Numbers divisible by both become *fizzbuzz*.

The Monad.Reader

In this tutorial, we focus on a single step of the game, that is to convert a natural number n into *fizz*, *buzz*, *fizzbuzz*, or its string representation.

There are a lot of solutions floating around the Internet, but most of them are, from our point of view, unsatisfactory. Exhibit A:

```
fizzbuzz :: Int → String
fizzbuzz n =
  if n `mod` 3 ≡ 0 ∧ n `mod` 5 ≡ 0 then
    "fizzbuzz"
  else if n `mod` 3 ≡ 0 then
    "fizz"
  else if n `mod` 5 ≡ 0 then
    "buzz"
  else
    show n
```

Exhibit B:

```
fizzbuzz :: Int → String
fizzbuzz n =
  if n `mod` 3 ≡ 0
  then "fizz" ++ if n `mod` 5 ≡ 0
                  then "buzz"
                  else ""
  else if n `mod` 5 ≡ 0
  then "buzz"
  else show n
```

Though both programs are correct with respect to the specification, Exhibit A, in some cases, performs the *'mod'3* and *'mod'5* tests more than once, while Exhibit B disperses the *buzzing* code into more than one place in the program. Meanwhile, we want there to be at most one place that outputs *fizz*, *buzz*, or the string representation, and each test to be performed only once. Outputting *fizzbuzz* should be done by executing the (one and only) piece of code that outputs *fizz*, followed by the (one and only) piece that outputs *buzz*. That is because *fizzing* and *buzzing* are two separate activities – consider the FizzBuzzHissHowl problem, where *hiss* and *howl* are printed for multiples of 7 and 11 respectively. The program design of Exhibit A and B would lead to an explosion of code complexity.

We can examine the output itself, which gives us an opportunity for yet another unsatisfactory attempt, Exhibit C:

```
(◁) :: String → String → String
"" ◁ s = s
```

```

a < s = a
fizzbuzz :: Int → String
fizzbuzz n = (( if n `mod` 3 ≡ 0 then "fizz" else "" )
              ++ if n `mod` 5 ≡ 0 then "buzz" else "" )
              < show n

```

The problem with this solution is far more subtle: many might say this solution is simple and elegant. Be that as it may, we do not like the fact that the `<` operator has to check if its first argument is empty. After all, we have already checked the conditions `'mod'3` and `'mod'5`, so the third test (`<`'s pattern matching) seems redundant from the information-flow point of view (compare Exhibit B, which always performs only two tests).

So, is out there a program that reflects the information-flow structure as Exhibit B, but, at the same time, is as modular as Exhibit C? Let's find out!

Skip-Halt-Print and contexts

If one feels overwhelmed by the number of (better or worse) possible ways to solve such a simple problem as FizzBuzz in Haskell, they can start with a simpler language. The one we propose is called Skip-Halt-Print, and it is very imperative.

A program in Skip-Halt-Print is a (possibly empty) list of commands, which are executed sequentially. There are only three different commands:

- ▶ SKIP is an idle instruction; it does nothing at all;
- ▶ HALT stops the computation; the rest of the program is not executed;
- ▶ PRINT `s` prints out the string `s`.

More formally, the syntax is given by the following grammar, where `c` denotes commands, `p` denotes programs, and `ε` is the empty program:

$$\begin{aligned}
 c &::= \text{SKIP} \mid \text{HALT} \mid \text{PRINT } \mathbf{s} \\
 p &::= c; p \mid \epsilon
 \end{aligned}$$

For brevity, we will give `s` also as an integer literal with implicit conversion. For example, the command `PRINT 42` is meant to print out the string of characters `42`.

The formal semantics can be given by a denotation function $\llbracket - \rrbracket : p \rightarrow \text{String}$, where *String* is the set of all strings of characters. In the following, `++` denotes concatenation and `""` denotes the empty string.

$$\begin{aligned}
 \llbracket \text{SKIP}; p \rrbracket &= \llbracket p \rrbracket \\
 \llbracket \text{HALT}; p \rrbracket &= \llbracket \epsilon \rrbracket = "" \\
 \llbracket \text{PRINT } \mathbf{s}; p \rrbracket &= \mathbf{s} ++ \llbracket p \rrbracket
 \end{aligned}$$

For example:

$$\begin{aligned} \llbracket \text{PRINT "studio"; SKIP; PRINT 54} \rrbracket &= \text{studio54} \\ \llbracket \text{PRINT "nuts"; HALT; PRINT "and bolts"} \rrbracket &= \text{nuts} \end{aligned}$$

For every natural number n , we construct a Skip-Halt-Print program that solves FizzBuzz for n . The building blocks for this construction are called *contexts* – they are programs with holes (one hole per context). We denote contexts by putting angle brackets $\langle - \rangle$ around the programs, while holes are denoted by the \bullet symbol. For example:

$$\langle \text{PRINT "keep"; } \bullet; \text{ PRINT "calm"} \rangle$$

A hole is a place in which we can stick another program and get a new program as a result. We denote this operation by juxtaposition:

$$\begin{aligned} &\langle \text{PRINT "keep"; } \bullet; \text{ PRINT "calm"} \rangle \langle \text{PRINT "nervous and never"} \rangle \\ &= \text{PRINT "keep"; PRINT "nervous and never"; PRINT "calm"} \end{aligned}$$

Two contexts can be composed, and for that we use the \circ symbol:

$$\begin{aligned} &\langle \text{SKIP; } \bullet; \text{ PRINT 0} \rangle \circ \langle \text{HALT; } \bullet \rangle \\ &= \langle \text{SKIP; HALT; } \bullet; \text{ PRINT 0} \rangle \end{aligned}$$

What does the FizzBuzz program do? Essentially, it prints out n , unless something else (like *fizzing* or *buzzing*) happens. This behaviour is captured by the following context:

$$\text{base}(n) = \langle \bullet; \text{ PRINT } n \rangle$$

What about *fizzing*? If only n is divisible by 3, it prints out *fizz*, but it also needs to prevent the default action from happening by HALT-ing the computation. In between PRINT-ing and HALT-ing anything (like *buzzing*) can happen:

$$\text{fizz}(n) = \begin{cases} \langle \text{PRINT "fizz"; } \bullet; \text{ HALT} \rangle & \text{if } n \bmod 3 = 0 \\ \langle \bullet \rangle & \text{otherwise} \end{cases}$$

The context for *buzzing* is analogous:

$$\text{buzz}(n) = \begin{cases} \langle \text{PRINT "buzz"; } \bullet; \text{ HALT} \rangle & \text{if } n \bmod 5 = 0 \\ \langle \bullet \rangle & \text{otherwise} \end{cases}$$

The program that solves FizzBuzz for n , which we call $fb(n)$, is a composition of all these contexts, which we cork with SKIP:

$$fb(n) = (base(n) \circ fizz(n) \circ buzz(n)) \text{ SKIP}$$

Examples:

$fb(1) = \text{SKIP}; \text{PRINT } 1$

$fb(3) = \text{PRINT "fizz"}; \text{SKIP}; \text{HALT}; \text{PRINT } 3$

$fb(5) = \text{PRINT "buzz"}; \text{SKIP}; \text{HALT}; \text{PRINT } 5$

$fb(15) = \text{PRINT "fizz"}; \text{PRINT "buzz"}; \text{SKIP}; \text{HALT}; \text{HALT}; \text{PRINT } 15$

Exercise 1. For any natural number n , give a Skip-Halt-Print program that solves the FizzBuzzHissHowl problem for n .

Exercise 2. What is the formal definition of the operations on contexts described above? Show that contexts with composition form a monoid; that is, \circ is associative, $(f \circ g) \circ h = f \circ (g \circ h)$, and $\langle \bullet \rangle$ is its left and right unit, $\langle \bullet \rangle \circ f = f$ and $f \circ \langle \bullet \rangle = f$ respectively.

Haskell implementation

Now, to solve FizzBuzz in Haskell, we implement Skip-Halt-Print, both syntax and semantics, together with the language of contexts. For each n , we construct the right composition of contexts as described above and then execute the resulting program.

Then, we apply a series of algebraic transformations that simplify the code into our proposed solution. By “algebraic”, we mean transformations that depend only on local properties of the components, without the actual understanding of the implemented algorithm. In other words, something that can be deduced solely from the shape of the code, like the *fold* pattern, and applied by simple equational calculation.

Direct definition

The commands of Skip-Halt-Print are implemented as a three-constructor data type *Cmd*, and the program is, of course, a list of commands. We call the $\llbracket - \rrbracket$ function *interp*.

```
data Cmd    = Skip | Halt | Print String
type Program = [Cmd]
```

The Monad.Reader

```
interp :: Program → String
interp (Skip    : xs) = interp xs
interp (Halt    : xs) = ""
interp (Print s : xs) = s ++ interp xs
interp []       = ""
```

Contexts are more tricky. Instead of specifying their syntax and operations, we encode them as functions from programs to programs (this technique is sometimes called *higher-order abstract syntax*). In this case, sticking the program in a context becomes Haskell's function application, and the composition of contexts becomes simply Haskell's \circ . However, note that not every Haskell function of the type $Program \rightarrow Program$ is a valid context in the sense specified in the previous section.

```
type Cont = Program → Program
fizz, buzz, base :: Int → Cont
fizz n | n `mod` 3 == 0 = λx → [Print "fizz"] ++ x ++ [Halt]
      | otherwise      = id
buzz n | n `mod` 5 == 0 = λx → [Print "buzz"] ++ x ++ [Halt]
      | otherwise      = id
base n                  = λx → x ++ [Print (show n)]
fb :: Int → Program
fb n = (base n ∘ fizz n ∘ buzz n) [Skip]
fizzbuzz :: Int → String
fizzbuzz n = interp (fb n)
```

Interpretation is a fold

To solve FizzBuzz for n , we first build a program (a datastructure) and then interpret it (by traversing the datastructure). This calls for some deforestation – the removal of the intermediate structures! First, we notice a known pattern here: *interp* is a fold. We can rewrite it as follows:

```
step :: Cmd → String → String
step Skip    t = t
step Halt    t = ""
step (Print s) t = s ++ t
interp = foldr step ""
```

Additionally, *foldr* has the following property (see Exercise 3):

$$\text{foldr } \textit{step} \text{ "" } p = \text{foldr } (\circ) \textit{id} (\textit{fmap } \textit{step} \text{ } p) \text{ ""}$$

So, instead of writing programs like

$$[\textit{Skip}, \textit{Halt}, \textit{Print} \text{ "c"}]$$

and interpreting them by folding with *step*, we can write programs like

$$[\textit{step } \textit{Skip}, \textit{step } \textit{Halt}, \textit{step } (\textit{Print} \text{ "c"})]$$

and interpret them by folding \circ . Also, we can inline the definition of *step*:

$$\begin{aligned} & [\textit{step } \textit{Skip}, \textit{step } \textit{Halt}, \textit{step } (\textit{Print} \text{ "c"})] \\ &= [\lambda t \rightarrow t, \lambda t \rightarrow \text{""}, \lambda t \rightarrow \text{"c"} \text{ ++ } t] \\ &= [\textit{id}, \textit{const} \text{ ""}, (\text{"c"} \text{ ++})] \end{aligned}$$

Why build and then interpret? We can manually deforest the situation by fusing the two: instead of

$$\text{foldr } (\circ) \textit{id} [\textit{id}, \textit{const}, (\text{"c"} \text{ ++})]$$

we write

$$\textit{id} \circ \textit{const} \circ (\text{"c"} \text{ ++})$$

In summary, we can define the next version of Skip-Halt-Print commands as follows:

```
type Program = String → String
skip, halt :: Program
skip = id
halt = const ""
print     :: String → Program
print = (+)
```

Now, our programs look like this:

$$\textit{print} \text{ "hello"} \circ \textit{skip} \circ \textit{print} \text{ "world"} \circ \textit{halt}$$

To execute them, we apply them to an empty string, for example:

$$(\textit{print} \text{ "hello"} \circ \textit{skip} \circ \textit{print} \text{ "world"} \circ \textit{halt}) \text{ ""} = \text{"helloworld"}$$

We need to accordingly adjust the bodies of our contexts:

```

type Cont = Program → Program
fizz, buzz, base :: Int → Cont
fizz n | n `mod` 3 ≡ 0 = λx → print "fizz" ∘ x ∘ halt
      | otherwise     = id
buzz n | n `mod` 5 ≡ 0 = λx → print "buzz" ∘ x ∘ halt
      | otherwise     = id
base n                = λx → x ∘ print (show n)
fizzbuzz :: Int → String
fizzbuzz n = (base n ∘ fizz n ∘ buzz n) skip ""

```

Notice that \circ is now overloaded: it composes both programs from commands (as in the bodies of functions *fizz*, *buzz*, and *base*) and contexts (as in the body of *fizzbuzz*).

Inlining

The truth is that we do not need to implement the entire Skip-Halt-Print language to solve FizzBuzz – our three contexts suffice. Thus, we inline the definitions of *base*, *skip*, *halt*, and *print* in *fizzbuzz*. We also put *fizz* and *buzz* as local definitions, so that we don't have to pass *n* around:

```

fizzbuzz :: Int → String
fizzbuzz n = (fizz ∘ buzz) id (show n)
where
  fizz | n `mod` 3 ≡ 0 = λx → const ("fizz" ++ x "")
      | otherwise     = id
  buzz | n `mod` 5 ≡ 0 = λx → const ("buzz" ++ x "")
      | otherwise     = id

```

Final polishing

As the last step, we abstract over the divisor and the printed message in *fizz* and *buzz*:

```

fizzbuzz :: Int → String
fizzbuzz n = (test 3 "fizz" ∘ test 5 "buzz") id (show n)
where
  test d s x | n `mod` d ≡ 0 = const (s ++ x "")
            | otherwise     = x

```

What is going on in this program? The (higher-order) function *test* has the following, longish type:

there is a small point it wants to convey: **Functional programmers! Remember higher-order functions!** They are your tool to express programs with non-trivial structure, to follow the information-flow more closely, to dynamically build your programs in runtime. A harsh, cantankerous functional programming pedagogue might say that they are such a basic tool that the final FizzBuzz program shouldn't appear complicated at all (and could be easily written by hand) if one knows their paradigm.

Closing remarks

The Skip-Halt-Print language is based on Edsger W. Dijkstra's Skip-Abort, which can be found in his textbook *A Discipline of Programming* [2, Chapter 4] (I am grateful to Tomasz Wierzbicki for the reference.) However, one needs to be aware of a difference in semantics between HALT and ABORT: the former peacefully ends the computation (like `return` in C-like languages), while the latter atrociously breaks it (like Haskell's `error` function).

I would also like to thank Jeremy Gibbons for his comments. The first idea for this tutorial sparkled in my head after Laurence E. Day's Facebook post:

I can write doctoral level Haskell without so much as missing a beat,
but I'd have a genuinely hard time writing a FizzBuzz program in Java.

The main point of this tutorial is that FizzBuzz is **not at all trivial**, but, many thanks to higher-order functions in Haskell, solvable. I don't know about Java.

References

- [1] Wikipedia. http://en.wikipedia.org/wiki/Fizz_buzz.
- [2] Edsger W. Dijkstra. **A discipline of programming**. Prentice-Hall series in automatic computation, Prentice-Hall, Incorporated (1976).